Example of DAE (index reduction with Baumgarte stabilization)

```
> with(plots):
```

The DAE (pendulum in cartesian coordinate)

```
> EQ1 := diff(x(t),t)-u(t) ;
  EQ2 := diff(y(t),t)-v(t) ;
  EQ3 := m*diff(u(t),t)+x(t)*lambda(t) ;
  EQ4 := m*diff(v(t),t)+y(t)*lambda(t)+m*g ;
  ALG := x(t)^2+y(t)^2-1 ; # algebraic contraint for the Pendulum
```

$$EQ1 := \frac{d}{dt} x(t) - u(t)$$

$$EQ2 := \frac{d}{dt} y(t) - v(t)$$

$$EQ3 := m \left(\frac{d}{dt} u(t) \right) + x(t)\, \lambda(t)$$

$$EQ4 := m \left(\frac{d}{dt} v(t) \right) + y(t)\, \lambda(t) + m\, g$$

$$ALG := x(t)^2 + y(t)^2 - 1 \quad (1)$$

```
> SUBS_RULE := solve( {EQ||(1..4)}, diff({x(t),y(t),u(t),v(t)},t) ) ;
```

$$SUBS\_RULE := \left\{ \frac{d}{dt} u(t) = -\frac{x(t)\, \lambda(t)}{m}, \frac{d}{dt} v(t) = -\frac{y(t)\, \lambda(t) + m\, g}{m}, \frac{d}{dt} x(t) = u(t), \frac{d}{dt} y(t) = v(t) \right\} \quad (2)$$

```
> ALG1 := diff(ALG,t) ;
  ALG1 := subs( SUBS_RULE, ALG1) ;
  ALG2 := diff(ALG1,t) ;
  ALG2 := collect(subs( SUBS_RULE, ALG2),lambda) ;
  #ALG2 := collect(simplify(subs(x(t)^2=1-y(t)^2,ALG2)),[lambda,m]);
```

$$ALG1 := 2\, x(t) \left(\frac{d}{dt} x(t) \right) + 2\, y(t) \left(\frac{d}{dt} y(t) \right)$$

$$ALG1 := 2\, x(t)\, u(t) + 2\, y(t)\, v(t)$$

$$ALG2 := 2 \left(\frac{d}{dt} x(t) \right) u(t) + 2\, x(t) \left(\frac{d}{dt} u(t) \right) + 2 \left(\frac{d}{dt} y(t) \right) v(t) + 2\, y(t) \left(\frac{d}{dt} v(t) \right)$$

$$ALG2 := \left(-\frac{2\, x(t)^2}{m} - \frac{2\, y(t)^2}{m} \right) \lambda(t) + 2\, u(t)^2 + 2\, v(t)^2 - 2\, y(t)\, g \quad (3)$$

Use Baumgarte stabilization for the constraint ALG = ALG''+2*zeta*omega*ALG'+omega^2ALG

```
> STAB_ALG := simplify(ALG2 + 2*zeta*omega*ALG1+omega^2*ALG) ;
```

$$STAB\_ALG := \frac{1}{m} \left(4\, x(t)\, u(t)\, m\, \omega\, \zeta + x(t)^2\, m\, \omega^2 + 4\, y(t)\, v(t)\, m\, \omega\, \zeta + y(t)^2\, m\, \omega^2 + 2\, u(t)^2\, m - 2\, x(t)^2\, \lambda(t) - 2\, y(t)^2\, \lambda(t) + 2\, v(t)^2\, m - 2\, y(t)\, g\, m - \omega^2\, m \right) \quad (4)$$

```
> LAMBDA_STAB := solve( subs(SUBS_RULE,STAB_ALG), {lambda(t)} ) ;
```

(5)

$$LAMBDA\_STAB := \left\{\lambda(t) = \frac{1}{2}\,\frac{1}{x(t)^2+y(t)^2}\Big(m\,\big(4\,u(t)\,x(t)\,\omega\,\zeta + x(t)^2\,\omega^2 + 4\,v(t)\,y(t)\,\omega\,\zeta + y(t)^2\,\omega^2 + 2\,u(t)^2 + 2\,v(t)^2 - 2\,y(t)\,g - \omega^2\big)\Big)\right\} \quad \textbf{(5)}$$

Diffentiate the stabilized constraint

```
> EQ5 := diff(lambda(t),t) -
         simplify(subs(SUBS_RULE,diff(subs(LAMBDA_STAB,lambda(t)),t))
  ) ;
```

$$EQ5 := \frac{d}{dt}\,\lambda(t) + \frac{1}{\left(x(t)^2+y(t)^2\right)^2}\Big(2\,x(t)^4\,\lambda(t)\,\omega\,\zeta + 4\,x(t)^2\,\lambda(t)\,y(t)^2\,\omega\,\zeta + 2\,x(t)^2\,u(t)^2\,m\,\omega\,\zeta + 2\,x(t)^2\,y(t)\,g\,m\,\omega\,\zeta - 2\,x(t)^2\,v(t)^2\,m\,\omega\,\zeta + 8\,x(t)\,u(t)\,y(t)\,v(t)\,m\,\omega\,\zeta + 2\,\lambda(t)\,y(t)^4\,\omega\,\zeta - 2\,u(t)^2\,y(t)^2\,m\,\omega\,\zeta + 2\,y(t)^3\,g\,m\,\omega\,\zeta + 2\,y(t)^2\,v(t)^2\,m\,\omega\,\zeta + 2\,x(t)^3\,\lambda(t)\,u(t) + 2\,x(t)^2\,\lambda(t)\,y(t)\,v(t) + 3\,x(t)^2\,v(t)\,g\,m + 2\,x(t)\,\lambda(t)\,u(t)\,y(t)^2 + 2\,x(t)\,u(t)^3\,m - 2\,x(t)\,u(t)\,y(t)\,g\,m + 2\,x(t)\,u(t)\,v(t)^2\,m - x(t)\,m\,\omega^2\,u(t) + 2\,\lambda(t)\,y(t)^3\,v(t) + 2\,u(t)^2\,y(t)\,v(t)\,m + y(t)^2\,v(t)\,g\,m + 2\,y(t)\,v(t)^3\,m - y(t)\,m\,\omega^2\,v(t)\Big) \quad \textbf{(6)}$$

```
> EQLAMBDA := solve(ALG2,{lambda(t)} ) ;
```

$$EQLAMBDA := \left\{\lambda(t) = \frac{m\,\left(u(t)^2 + v(t)^2 - y(t)\,g\right)}{x(t)^2 + y(t)^2}\right\} \quad \textbf{(7)}$$

```
> INI := x(0)=1,y(0)=0,u(0)=0,v(0)=0 ;
  INI := INI, lambda(0)=1 ; #subs(INI,lambda(0)=subs(t=0,subs
  (EQLAMBDA,lambda(t)))) ;
```

$$INI := x(0) = 1,\ y(0) = 0,\ u(0) = 0,\ v(0) = 0$$

$$INI := x(0) = 1,\ y(0) = 0,\ u(0) = 0,\ v(0) = 0,\ \lambda(0) = 1 \quad \textbf{(8)}$$

```
> SUBS_VALUES := g=9.81, m=1, zeta=0.5, omega=5 ;
```

$$SUBS\_VALUES := g = 9.81,\ m = 1,\ \zeta = 0.5,\ \omega = 5 \quad \textbf{(9)}$$

```
> SOL := dsolve( subs(SUBS_VALUES,{EQ||(1..5),INI}),type=numeric ) ;
```

$SOL :=$ **proc**(x_rkf45) ... **end proc** **(10)**

```
> odeplot(SOL, [x(t),y(t)], 0..50, numpoints=1000) ;
```



Use stabized constraints substituted in the ODE part

```
> LAMBDA_STAB ;
```

$$\left\{\lambda(t) = \frac{1}{2}\,\frac{1}{x(t)^2 + y(t)^2}\left(m\left(4\,u(t)\,x(t)\,\omega\,\zeta + x(t)^2\,\omega^2 + 4\,v(t)\,y(t)\,\omega\,\zeta + y(t)^2\,\omega^2 + 2\,u(t)^2 + 2\,v(t)^2 - 2\,y(t)\,g - \omega^2\right)\right)\right\} \tag{11}$$

```
> EQ_STAB1 := subs(LAMBDA_STAB,EQ1) ;
  EQ_STAB2 := subs(LAMBDA_STAB,EQ2) ;
  EQ_STAB3 := subs(LAMBDA_STAB,EQ3) ;
  EQ_STAB4 := subs(LAMBDA_STAB,EQ4) ;
```

$$EQ\_STAB1 := \frac{d}{dt}\,x(t) - u(t)$$

$$EQ\_STAB2 := \frac{d}{dt}\,y(t) - v(t)$$

$$EQ\_STAB3 := m\left(\frac{d}{dt}\,u(t)\right) + \frac{1}{2}\,\frac{1}{x(t)^2 + y(t)^2}\left(x(t)\,m\left(4\,u(t)\,x(t)\,\omega\,\zeta + x(t)^2\,\omega^2 + 4\,v(t)\,y(t)\,\omega\,\zeta + y(t)^2\,\omega^2 + 2\,u(t)^2 + 2\,v(t)^2 - 2\,y(t)\,g - \omega^2\right)\right)$$

$$EQ\_STAB4 := m \left(\frac{d}{dt} v(t)\right) + \frac{1}{2} \frac{1}{x(t)^2 + y(t)^2} \left(y(t)\, m \left(4\, u(t)\, x(t)\, \omega\, \zeta + x(t)^2\, \omega^2 + 4\, v(t)\, y(t)\, \omega\, \zeta + y(t)^2\, \omega^2 + 2\, u(t)^2 + 2\, v(t)^2 - 2\, y(t)\, g - \omega^2\right)\right) + m\, g \quad \textbf{(12)}$$

```
> INI := x(0)=1,y(0)=0,u(0)=0,v(0)=0 ;
```

$$INI := x(0) = 1, y(0) = 0, u(0) = 0, v(0) = 0 \quad \textbf{(13)}$$

```
> SUBS_VALUES := g=9.81, m=1, zeta=0.5, omega=1 ;
```

$$SUBS\_VALUES := g = 9.81, m = 1, \zeta = 0.5, \omega = 1 \quad \textbf{(14)}$$

```
> SOL := dsolve( subs(SUBS_VALUES,{EQ_STAB||(1..4),INI}),type=numeric
) ;
```

$SOL :=$ **proc**(x_rkf45) ... **end proc** **(15)**

```
> odeplot(SOL, [x(t),y(t)], 0..135, numpoints=10000) ;
```